UFR-MIG Mécanique, 2006 AB
Dynamique des Vibrations linéaires
Poutre en mouvement de flexion
Poutre Encastree-Encastree

```
> restart;with(plots):with(LinearAlgebra):unprotect(D):
> Mat:=Matrix(4,4):Vect:=Vector(4,[C ,D, E, F]):
> PHI:=C*sin(beta*x)+D*cos(beta*x)+E*sinh(beta*x)+F*cosh(beta*x):
> Pente:=diff(PHI,x):
> Moment_F:=diff(PHI,x,x):
> Effort_T:=diff(PHI,x,x,x):
> temp:=subs(x=0,PHI):
> Mat[1,1]:=coeff(temp,C):Mat[1,2]:=coeff(temp,D):Mat[1,3]:=coeff(
  temp,E):Mat[1,4]:=coeff(temp,F):
> temp:=subs(x=0,Pente):
> Mat[2,1]:=coeff(temp,C):Mat[2,2]:=coeff(temp,D):Mat[2,3]:=coeff(
  temp,E):Mat[2,4]:=coeff(temp,F):
> temp:=subs(x=L,PHI):
> Mat[3,1]:=coeff(temp,C):Mat[3,2]:=coeff(temp,D):Mat[3,3]:=coeff(
  temp,E):Mat[3,4]:=coeff(temp,F):
> temp:=subs(x=L,Pente):
> Mat[4,1]:=coeff(temp,C):Mat[4,2]:=coeff(temp,D):Mat[4,3]:=coeff(
  temp,E):Mat[4,4]:=coeff(temp,F):
> #print(Mat):
> Prod:=(Mat.Vect):
> simplify(Determinant(Mat)/beta^2/2=0):
> sol1:=solve({Prod[1],Prod[2]},{E,F}):
> PHI:=collect(collect(collect(collect(subs(sol1,PHI),F),E),D),C):
> Pente:=collect(collect(collect(collect(subs(sol1,Pente),F),E),D)
  ,C):
> sol1:=solve({subs(x=L,PHI)},C):
> Phi1:=subs(sol1,PHI):
> Phi1:=collect(Phi1,D);
```

$$\Phi 1 := \left(- \frac{(\sin(\beta\, x) - \sinh(\beta\, x))\,(-\cos(\beta\, L) + \cosh(\beta\, L))}{-\sin(\beta\, L) + \sinh(\beta\, L)} + \cos(\beta\, x) - \cosh(\beta\, x) \right) \mathrm{D}$$

```
> sol2:=solve({subs(x=L,Pente)},D):
> Phi2:=subs(sol2,PHI):
> Phi2:=collect(Phi2,C);
```

$$\Phi 2 := \left(\sin(\beta\, x) - \sinh(\beta\, x) - \frac{(\cos(\beta\, x) - \cosh(\beta\, x))\,(-\cos(\beta\, L) + \cosh(\beta\, L))}{\sin(\beta\, L) + \sinh(\beta\, L)} \right) C$$

```
> beta:=sqrt(22.37);
  L:=1;
> tr1:=plot({subs(D=1,Phi1)},x=0..1):
  tr2:=plot({subs(C=1,Phi2)},x=0..1):

> display({tr1,tr2});
```

$$\beta := 4.729693436$$

$$L := 1$$




```
> animate(plot,[subs(D=1,Phi1)*sin(t),x=0..1],t=0..2*Pi*23/24,fram
  es=24,thickness=5);
```




```
> animate(plot,[subs(C=1,Phi2)*sin(t),x=0..1],t=0..2*Pi*23/24,fram
  es=24,thickness=5);
```




```
> #  Deformee supposee
  unassign('L');
  dep_R:=C*f(t)*(1-cos(2*Pi*x/L));
  tr3:=plot({subs(C=1,L=1,C*(1-cos(2*Pi*x/L)))},x=0..1):
  display(tr3);
```

$$dep\_R := C\, \mathrm{f}(t) \left(1 - \cos\left(\frac{2\,\pi\,x}{L}\right)\right)$$




```
> #  Energie de deformation
```

```
temp:=diff(dep_R,x,x)^2:
> UU:=E*II*int(temp,x=0..L)/2;
```

$$UU := \frac{4\, E\, II\, \pi^4\, C^2\, \mathrm{f}(t)^2}{L^3}$$

```
#  Energie cinetique poutre
> temp:=diff(dep_R,t)^2:
> TT:=rho*S*int(temp,x=0..L)/2;
```

$$TT := \frac{3}{4}\, \rho\, S\, L\, C^2 \left(\frac{d}{dt}\, \mathrm{f}(t)\right)^2$$

```
#  Equation de Lagrange pour la poutre
> KK:=diff(subs(f(t)=f_t,UU),f_t);
  k:=coeff(subs(C=1,KK),f_t);
> MM:=diff(subs(d_f_t=diff(f(t),t),diff(subs(diff(f(t),t)=d_f_t,TT
  ),d_f_t)),t);
> m:=coeff(subs(C=1,MM),diff(f(t),`$`(t,2)));
```

$$KK := \frac{8\, E\, II\, \pi^4\, C^2\, f\_t}{L^3}$$

$$k := \frac{8\, E\, II\, \pi^4}{L^3}$$

$$MM := \frac{3}{2}\, \rho\, S\, L\, C^2 \left(\frac{d^2}{dt^2}\, \mathrm{f}(t)\right)$$

$$m := \frac{3\, \rho\, S\, L}{2}$$

```
#  Equation du mouvement de la poutre (1ddl)
> EOM:=MM+KK;
```

$$EOM := \frac{3}{2}\, \rho\, S\, L\, C^2 \left(\frac{d^2}{dt^2}\, \mathrm{f}(t)\right) + \frac{8\, E\, II\, \pi^4\, C^2\, f\_t}{L^3}$$

```
# Pulsation approchee
> omega_1:=(sqrt(k/m));
  omega_1:=evalf(%);
  omega_Th:=22.37*(E*II/L^4/rho/S)^(1/2);
  delta_percent:=(omega_1-omega_Th)/omega_Th;
```

$$omega\_1 := \frac{4\sqrt{3}\, \pi^2 \sqrt{\frac{E\, II}{L^4\, \rho\, S}}}{3}$$

$$omega\_1 := 22.79287503 \sqrt{\frac{E\, II}{L^4\, \rho\, S}}$$

$$omega\_Th := 22.37 \sqrt{\frac{E\, II}{L^4\, \rho\, S}}$$

$$delta\_percent := 0.01890366696$$

```
#
#  Cas de la masse;
#  Energie cinetique
> temp1:=subs(x=2*L/3,dep_R);
  temp2:=diff(temp1,t);
  delta_TT1:=(rho*S*L/20)*temp2^2/2;
```

$$temp1 := C\, \mathrm{f}(t)\left(1 - \cos\left(\frac{4\,\pi}{3}\right)\right)$$

$$temp2 := \frac{3}{2}\, C\left(\frac{d}{dt}\, \mathrm{f}(t)\right)$$

$$delta\_TT1 := \frac{9}{160}\, \rho\, S\, L\, C^2 \left(\frac{d}{dt}\, \mathrm{f}(t)\right)^2$$

```
#  Equation de Lagrange (cinetique)
> temp3:=subs(diff(f(t),t)=f_t,delta_TT1):
  temp4:=diff(temp3,f_t):
  temp5:=subs(f_t=diff(f(t),t),temp4):
  temp6:=diff(temp5,t):
  delta_m1:=coeff(subs(C=1,temp6),diff(f(t),`$`(t,2)));
```

$$delta\_m1 := \frac{9\, \rho\, S\, L}{80}$$

```
> omega_1m:=(sqrt(k/(m+delta_m1)));
  omega_1m:=evalf(%);
  omega_Th:=22.37*(E*II/L^4/rho/S)^(1/2);
  delta_percent:=evalf(omega_1m-omega_Th)/omega_Th;
```

$$omega\_1m := \frac{8\, \sqrt{1290}\, \pi^2 \sqrt{\frac{E\, II}{L^4\, \rho\, S}}}{129}$$

$$omega\_1m := 21.98340076\, \sqrt{\frac{E\, II}{L^4\, \rho\, S}}$$

$$omega\_Th := 22.37\, \sqrt{\frac{E\, II}{L^4\, \rho\, S}}$$

$$delta\_percent := -0.01728204023$$

```
#
#  Cas du ressort
#  Energie de deformation
> temp1:=subs(x=2*L/3,dep_R);
  temp2:=eval((temp1)^2):
  Ke:=E*II/(L^3);
  delta_UU1:=(Ke)*temp2/2:
  temp3:=subs(f(t)=f_t,delta_UU1):
  temp4:=diff(temp3,f_t):
```

```
delta_k1:=subs(C=1,coeff(temp4,f_t));
```

$$temp1 := C\,\mathrm{f}(t)\left(1-\cos\left(\frac{4\,\pi}{3}\right)\right)$$

$$Ke := \frac{E\,II}{L^3}$$

$$delta\_k1 := \frac{9\,E\,II}{4\,L^3}$$

```
> omega_1R:=(sqrt(((k+delta_k1))/m));
  omega_1R:=evalf(%);
  omega_Th:=22.37*(E*II/L^4/rho/S)^(1/2);
  delta_percent:=evalf(omega_1R-omega_Th)/omega_Th;
```

$$omega\_1R := \frac{\sqrt{6}\,\sqrt{\frac{\frac{8\,E\,II\,\pi^4}{L^3}+\frac{9\,E\,II}{4\,L^3}}{\rho\,S\,L}}}{3}$$

$$omega\_1R := 22.82575634\,\sqrt{\frac{E\,II}{L^4\,\rho\,S}}$$

$$omega\_Th := 22.37\,\sqrt{\frac{E\,II}{L^4\,\rho\,S}}$$

$$delta\_percent := 0.02037355118$$

```
>
>
>
>
```